
Creation of a Normal Program

As a rule, after having coded a number of simple application programs in MQL4, a programmer goes to a more complex project - to creation of a convenient program for practical use. Simple programs, in some cases, do not satisfy the needs of a trading programmer for at least two reasons:

1. The functional boundedness of simple programs does not allow them to provide a trader with all necessary information and trading management tools, which doesn't make the use of such programs efficient enough.

2. The code imperfection of simple programs makes difficult their further development aimed at expanded services.

In this section, we represent one of the possible alternatives of realizing a trading Expert Advisor that can be considered as a basis for your own projects.

· Structure of a Normal Program
The availability of many user-defined functions in a program allows you to create powerful and flexible algorithms to process information. Compiler directive #include allows you to use your function (once written and debugged) in other programs. In this manner, you can create your own libraries or use open-source developments of other programmers. 

· Order Accounting 
The section considers an example of user-defined function Terminal() that is realized in a separate include file with the extension .mqh. Such files are connected to the program code during compilation using the directive #include.

· Data Function
An example of one more user-defined function that helps to organize the output of text information about the current work of an EA. This function allows you to abandon the function Comment() to display the text in the chart window. The function is realized as an indicator in a separate subwindow of the chart window.

· Event Tracking Function
A trader cannot always notice all the events during trading. The program written in MQL4 allows you to detect the changes in any trading conditions or situations. user-defined function Events() connects to the EA using the directive #include and applies calls to another include function, Inform(). 

· Volume Defining Function
The calculation of the volume of a position to be opened is one of the tasks of equity/risk management. The user-defined function Lot() is a small example used for these purposes.

· Trading Criteria Defining Function
The most important part of any trading is detection of the times of entering the market and that of closing a position. The creation of trading rules or criteria is the core of any Expert Advisor. User-defined function Criterion() is connected using the directive #include. It shows how an EA can decide on the basis of the indicator values about whether the current situation complies with one or another criterion.

· Trade Functions
The current situation has been analyzed with the function Criterion(), so now we have to make trades: open, close, modify or delete a pending order. All these operations can be put in separate user-defined functions: Trade(), Close_All() and Open_Ord(). The protecting stop orders are moved using the user-defined function Tral_Stop().

· Error Processing Function
Error control is an integral part of an Expert Advisor. This is you who determines how to process the message about the busy trade context, about no prices for the requested symbol, etc. In some cases, it is sufficient to display a message about the error. In other cases, it would be reasonable to try and repeat the trade request after a certain amount of time. It is necessary to determine how one or another error will be processed. The user-defined function Errors() shown in this section processes errors using the selection operator switch().

Structure of a Normal Program

The outstanding feature of a normal program is its structure that allows your to easily use some user-defined functions or other. For convenience, user-defined functions are usually formed as include files (.mqh) that are stored in the folder Terminal_directory\experts\include.

As a rule, a normal program contains all three special functions that call the necessary user-defined functions for execution. User-defined functions, in their turn, can call other user-defined functions for execution, each of them having its own functionally bounded purpose.

An Expert Advisor can contain user-defined functions with the most diverse properties. Some functions, for example, are intended for tracking events and convenient data output, other functions are used to form trade requests, the third functions are intended for various calculations, for example, for defining of trading criteria, calculation of order costs, etc. The decision on what functions to use in a program depends on the EA purpose and on what service the program is intended to provide the user with. In Fig. 155 below, you can see a block diagram of a normal trading EA built on a small set of user-defined functions.


The arrows in the diagram show the relationships between functions. For example, the order accounting function in the EA is called from the special functions init() and start(); it can also be called from any other location in the program. In the right part of the diagram, the connecting arrows are shown that symbolize the possibility to call one user-defined function from another. For example, the function defining trading criteria cannot be called from any special function, but it can be called for execution by a trade function. The data function is called from the special function deinit() and, if necessary, from other functions, for example, from the error processing function, trade functions, or from the event tracking function. The include file of variable declaration cannot be called from any function, since the code contained in it is not the description of a function that can be called and executed. This file contains the lines of global variable declaration, so it is just a part of an EA. In order to understand the way the different parts of an EA interrelate in, let's consider the order of creation and use of include files.

Using Include Files

If an application program contains a great variety of program lines, it is sometimes difficult to debug it - the programmer has to scroll the program text many times in order to make changes in the code in one or another location. In such cases, for convenience, the program can be divided into several fragments, each formed as a separated include file. Include files can contain any code fragments to be used in the program. Each function used in the program is usually formed as an include file. If several functions are logically interconnected, one include file may contain the description of several user-defined functions.

In the section Information about an Account, we consider an example of the protection code that prevents from unauthorized use of business programs. In the Expert Advisor check.mq4, the corresponding part of the program code is represented as the user-defined function Check(), the description of which is directly contained in the source code of the EA. In the EA usualexpert.mq4 (see Fig. 91.3), this function is used, too. However, in this case, the function is formed as the include file Check.mqh.

User-Defined Function Check()

bool Check()
The function returns TRUE, if the conditions of using the application program are complied with. Otherwise, it returns FALSE.

The conditions of using the program are considered as complied with, if:

· the program is used on a demo account; 

· the account is opened with SuperBank; 

· the user has set the correct value of the external variable Parol when working on a real account. 

The include file Check.mqh containing the description of the user-defined function Check():

//----------------------------------------------------------------------------------
// Check.mqh
// The program is intended to be used as an example in MQL4 Tutorial.
//----------------------------------------------------------------------------- 1 --
// The function checking legality of the program used
// Inputs:
// - global variable 'Parol'
// - local constant "SuperBank"
// Returned values:
// true  - if the conditions of use are met
// false - if the conditions of use are violated
//----------------------------------------------------------------------------- 2 --
extern int Parol=12345; // Password to work on a real account
//----------------------------------------------------------------------------- 3 --
bool Check()                           // User-defined unction
  {
   if (IsDemo()==true)                 // If it is a demo account, then..
      return(true);                    // .. there are no other limitations
   if (AccountCompany()=="SuperBank")  // For corporate clients..
      return(true);                    // ..no password is required
   int Key=AccountNumber()*2+1000001;  // Calculate the key 
   if (Parol==Key)                     // If the password is true, then..
      return(true);                    // ..allow the user to work on a real account
   Inform(14);                         // Message about unauthorized use
   return(false);                      // Exit the user-defined function
  }
//----------------------------------------------------------------------------- 4 --
It is easy to note that the name of the include file is the same as the name of the function it contains. This is not required by the rules of MQL4. Generally, the name of an include file is not required to be the same as the name of the function it contains. It becomes even clearer, if we consider that one file may contain the description of several functions or of a program fragment that is not a function at all. However, the practice of giving an include file the same name as that of the contained function is feasible. This considerably facilitates the work of a programmer: using the file names, he or she will easily know what functions there are in the folder Terminal_directory\experts\include. In order to include the program fragment contained in the file, you should use the directive #include.

Directive #include

	#include <File name>
#include "File name"


The directive #include can be specified in any location in the program. However, all included fragments are placed at the beginning of the source text file. Preprocessor will replace the line #include <file_name> (or #include "file_name") with the contents of the file of the given name.

Angle brackets mean that the file will be taken from the standard directory Terminal_directory\experts\include (the current directory is not viewed). If the file name is enclosed in quotation marks, it will be searched in the current directory, namely, in the directory that contains the basic file of the source text (the standard directory is not viewed).

Below is a normal Expert Advisor, usualexpert.mq4. All include files are placed in the head part of the program.

//----------------------------------------------------------------------------------------
// usualexpert.mq4
// The code should be used for educational purpose only.
//----------------------------------------------------------------------------------- 1 --
#property copyright "Copyright © Book, 2007"
#property link      "http://AutoGraf.dp.ua"
//----------------------------------------------------------------------------------- 2 --
#include <stdlib.mqh>
#include <stderror.mqh>
#include <WinUser32.mqh>
//----------------------------------------------------------------------------------- 3 --
#include <Variables.mqh>   // Description of variables 
#include <Check.mqh>       // Checking legality of programs used
#include <Terminal.mqh>    // Order accounting
#include <Events.mqh>      // Event tracking function
#include <Inform.mqh>      // Data function
#include <Trade.mqh>       // Trade function
#include <Open_Ord.mqh>    // Opening one order of the preset type
#include <Close_All.mqh>   // Closing all orders of the preset type
#include <Tral_Stop.mqh>   // StopLoss modification for all orders of the preset type
#include <Lot.mqh>         // Calculation of the amount of lots
#include <Criterion.mqh>   // Trading criteria
#include <Errors.mqh>      // Error processing function
//----------------------------------------------------------------------------------- 4 --
int init()                             // Special function 'init'
  {
   Level_old=MarketInfo(Symbol(),MODE_STOPLEVEL );//Min. distance
   Terminal();                         // Order accounting function 
   return;                             // Exit init() 
  }
//----------------------------------------------------------------------------------- 5 --
int start()                            // Special function 'start'
  {
   if(Check()==false)                  // If the usage conditions..
      return;                          // ..are not met, then exit
   PlaySound("tick.wav");              // At every tick
   Terminal();                         // Order accounting function 
   Events();                           // Information about events
   Trade(Criterion());                 // Trade function
   Inform(0);                          // To change the color of objects
   return;                             // Exit start()
  }
//----------------------------------------------------------------------------------- 6 --
int deinit()                           // Special function deinit()
  {
   Inform(-1);                         // To delete objects
   return;                             // Exit deinit()
  }
//----------------------------------------------------------------------------------- 7 --
In block 2-3, we included into the program standard files stdlib.mqh, stderror.mqh and WinUser32.mqh using the directive #include. It is not always necessary to use these files in the program. For example, file stderror.mqh contains the definition of standard constants used in error processing. If error processing is not provided in the program (i.e., these constants are not used), then there is no need to include this file into the source text. At the same time, it is usually necessary to include these files in a normal program.

In block 3-4, the program includes some files containing the description of user-defined functions. Using the directive #include in the line below:

#include <Check.mqh>       // Checking legality of programs used
the source text of the program includes the user-defined function Check(). Programmer can see the source code of the EA (in this case, usualexpert.mq4) as it is represented above. However, the source text of the program is modified at compilation, namely, each line containing the directive #include is replaced in the program with the text contained in the file of the given name. Thus, an executable file .ex4 is created on the basis of the full code of the Expert Advisor, in which each line #include<File name> (or #include "File name") is replaced with the corresponding code fragment.

An example of an include file containing a program fragment that is not the function description is file Variables.mqh. This file is included in the program text in the line:

#include <Variables.mqh>   // Description of variables
and contains the description of global variables used by different user-defined functions.

//----------------------------------------------------------------------------
// Variables.mqh
// The code should be used for educational purpose only.
//----------------------------------------------------------------------- 1 --
// Description of global variables
extern double Lots    = 0.0;// Amount of lots
extern int Percent    = 0;  // Allocated funds percentage
extern int StopLoss   =100; // StopLoss for new orders (in points) 
extern int TakeProfit =40;  // TakeProfit for new orders (in points)
extern int TralingStop=100; // TralingStop for market orders (in points)
//----------------------------------------------------------------------- 2 --
int 
   Level_new,           // New value of the minimum distance
   Level_old,           // Previous value of the minimum distance
   Mas_Tip[6];          // Order type array
                        // [] order type: 0=B,1=S,2=BL,3=SL,4=BS,5=SS
//----------------------------------------------------------------------- 3 --
double
   Lots_New,            // Amount of lots for new orders
   Mas_Ord_New[31][9],  // Current order array ..
   Mas_Ord_Old[31][9];  // .. old order array
                        // 1st index = order number in the list 
                        // [][0] cannot be detected
                        // [][1] order open price (abs. price value)
                        // [][2] StopLoss of the order (abs. price value)
                        // [][3] TakeProfit of the order (abs. price value)
                        // [][4] order number        
                        // [][5] order volume (abs. price value)
                        // [][6] order type 0=B,1=S,2=BL,3=SL,4=BS,5=SS
                        // [][7] Order magic number
                        // [][8] 0/1 the fact of availability of comments
//----------------------------------------------------------------------- 4 --
According to the rules of MQL4, any variable (including global ones) must be declared before the first reference to this variable. For this reason, the file Variables.mqh is included into the program and located above the files of functions that use the values of the variables specified in this file. For the same reason, all global variables are placed in this file.

In some (rare) cases, it is technically possible to declare a global variable in an include file that describes the function, in which the value of this variable is first used in the program. In such cases, it is necessary to keep the file including order. I.e., the program line containing the directive #include that includes the file (with the declaration of a global variable) into the program must be located above in the text than the lines that include other files using the value of this global variable.

In other cases, it is even technically impossible to do this. For example, if we have two include files, each of them using two global variables, one of which is declared in one file and the other is declared in the other file, then we will get an error when compiling such a program, since, regardless the file including order, one of the variables is used before it is declared in the program. This is why it is a usual practice in a normal program to declare all global variables, without any exceptions, in one file that is included in the program before other files containing the description of user-defined functions.

In block 1-2 of the include file Variables.mqh, all external variables are specified, the values of which determine the amount of lots for new orders, the free margin percentage allocated for new orders, the requested prices for stop orders of the market orders to be opened, as well as the distance of TralingStop for modification of StopLoss order. In blocks 2-4, other global variables are given, the sense of which will become clear upon considering the corresponding user-defined functions. The subsections of this section represent include files, each containing the description of the user-defined function of the same name.

Order Accounting

We mentioned above that there were no strict rules for making program algorithms. At the same time, the overwhelming majority of algorithms imply making one's trading decisions according to the current status of the orders available. In some cases, for example, opening a market order needs no other market orders available as of the moment of the trade. In other cases, no stop orders available on the market order can be a necessary condition for placing of a pending order. We also know some algorithms that imply placing two differently directed pending orders.

In order to have met the requirements of one or another tactic or strategy by the moment of decision making, you should know about the current status - what market and pending orders are available and what characteristics do they have? You may use one of the two possible solutions.

According to the first solution, the necessary program code fragment (in which the orders are analyzed) is written directly in the location in the program, where the available set of orders and their characteristics should be found out. This solution is technically feasible, but it turns out to be inefficient, if you want to make changes in the algorithm. In this case, the programmer has to analyzes all locations in the program where order statuses are analyzed, and make changes in each location. Another, more effective solution is to create a universal order accounting function once and use it every time, when you want to update the information about the current order statuses. On the one hand, this solution allows you to reduce the program code. On the other hand, it allows the programmer to use this ready-made function when coding other programs.

In order to create an order accounting function correctly, you should first decide what parameters must be accounted. In most cases, the values of the following parameters are used in making trade decisions:

· total amount of orders; 

· the amount of orders of each type (for example, the amount of Buy orders, SellStop orders, or BuyLimit orders, etc.); 

· all characteristics of each order (ticket, StopLoss and TakeProfit levels, volume in lots, etc.). 

The above information must be available to other functions, namely to those, in which this information is processed. For this reason, all parameters that characterize order statuses are the values of global arrays. Totally, three arrays are provided for order accounting:

· the array of current orders, Mas_Ord_New, that contains information about all characteristics of all market and pending orders available at the current moment, namely, within the period of the last execution of the function; 

· the array of old orders, Mas_Ord_Old, that contains information about all characteristics of all market and pending orders available at the moment of the preceding execution of the function; 

· the array Mas_Tip, the values of which are the amounts of orders of different types (at the current moment). 

Arrays Mas_Ord_New and Mas_Ord_Old are similar and equidimensional; the difference between them is that the former one reflects the current status of orders, whereas the latter one shows the preceding status. Let's give a closer consideration to the values contained in the elements of those arrays.

Table 4. Correspondence of the elements of arrays Mas_Ord_New and Mas_Ord_Old with order characteristics.

	
	Not defined
	Open Price
	StopLoss
	TakeProfit
	Order Number
	Volume,
in lots
	Order Type
	Magic Number
	Comment

	Indexes
	0
	1
	2
	3
	4
	5
	6
	7
	8

	0
	2.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0

	1
	0.0
	1.2583
	1.2600
	1.2550
	123456.0
	1.4
	1.0
	1177102416.0
	1.0

	2
	0.0
	1.2450
	1.2580
	1.2415
	123458.0
	2.5
	2.0
	1177103358.0
	0.0

	3
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0

	. . .
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0

	30
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0
	0.0


The first index of the array (lines) defines the number of the order in the array. The characteristics of the first order detected (among all opened market orders and placed pending orders) are placed in the first line of the array, those of the second order detected will be placed in the second line, etc. The array size for the first index is equal to 31, thus, the array is intended to store information about 30 orders at the most, if they are simultaneously available on one trading account. If the trading strategy allows the availability of more than thirty orders simultaneously, you should specify the corresponding value for the first index when you declare the array. (In most cases, the value of 30 considerably exceeds the real need usually ranging from 2 to 10-15 orders. We use the value of 30 in this example, because we suppose that the function can be used for very unusual trading strategies, as well).

The second index in the array (columns) represents order characteristics. Each element of the array with the second index equal to 1 contains the value of the order open price, with index 2 - they contain the value of StopLoss order, 3 - TakeProfit, etc. (see Table 4). Array element with index [0][0] has a value that is equal to the total amount of orders available in the array. No array elements having the first or the second indexes equal to 0 are used (except element having index [0][0]).

Table 4 represents the status of an array that contains information about two orders that are simultaneously available in trading at a certain moment. The array element Mas_Ord_New[0][0] has the value of 2.0 - the total amount of orders is two. The elements in the first line of the array contain the values of the characteristics of the market order Sell (Mas_Ord_New[1][6] = 1.0, see Types of Trades) opened for 1.4 lot (Mas_Ord_New[1][5] =1.4) and having the number 123456 (Mas_Ord_New[1][4] =123456.0). The value of the element Mas_Ord_New[1][8] =1.0 means that this order has non-empty comment field. In the second line of the array, the values that characterize the second order are contained. Particularly, the array element Mas_Ord_New[2][6] has the value of 2.0, it means that it is BuyLimit.

Array Mas_Tip represents the amount of orders of each type. The values of this array indexes are assigned to the types of trades (see Types of Trades). This means that the element of array Mas_Tip with index 0 contains the amount of market orders of the Buy type simultaneously available in trading, index 1 means the amount of Sell orders, index 2 means that of BuyLimit orders, etc. For the situation shown in Table 4, the elements of array Mas_Tip will have the following values:

Table 5. Correspondence of the elements of array Mas_Tip with the amount of orders of different types.

	
	Buy
	Sell
	BuyLimit
	SellLimit
	BuyStop
	SellStop

	Index
	0
	1
	2
	3
	4
	5

	Value
	0
	1
	1
	0
	0
	0


In this case, the values of the elements of array Mas_Tip imply the following: Mas_Tip[1] equal to 1 means that there is one Sell order traded; Mas_Tip[2] equal to 1 means that there is one pending order BuyLimit in trading. Other elements of the array have zero values - this means that there are no orders of those types in trading. If there are several orders of the same type simultaneously available in trading, the corresponding element of the array will have the value equal to the amount of such orders. For example, if there are three orders BuyStop in trading, the element Mas_Tip[4] will have the value of 3.

The order accounting function Terminal() suggested here is formed as include file Terminal.mqh.

User-Defined Function Terminal()

int Terminal()
The function accounts market and pending orders. The execution of the function results in changing of the values of the following global arrays:

· Mas_Ord_New - the array of characteristics of orders available as of the moment of the function execution; 

· Mas_Ord_Old - the array of characteristics of orders available as of the moment of the preceding execution of the function; 

· Mas_Tip - the array of the total amount of orders of all types. 

Include file Terminal.mqh that contains the description of the order accounting function Terminal():

//--------------------------------------------------------------------
// Terminal.mqh
// The code should be used for educational purpose only.
//------------------------------------------------------------------------------ 1 --
// Order accounting function
// Global variables:
// Mas_Ord_New[31][9]   // The latest known orders array
// Mas_Ord_Old[31][9]   // The preceding (old) orders array
                        // 1st index = order number 
                        // [][0] not defined
                        // [][1] order open price (abs. price value)
                        // [][2] StopLoss of the order (abs. price value)
                        // [][3] TakeProfit of the order (abs. price value)
                        // [][4] order number        
                        // [][5] order volume in lots (abs. price value)
                        // [][6] order type 0=B,1=S,2=BL,3=SL,4=BS,5=SS
                        // [][7] order magic number
                        // [][8] 0/1 comment availability
// Mas_Tip[6]           // Array of the amount of orders of all types
                        // [] order type: 0=B,1=S,2=BL,3=SL,4=BS,5=SS
//------------------------------------------------------------------------------ 2 --
int Terminal()
  {
   int Qnt=0;                          // Orders counter
 
//------------------------------------------------------------------------------ 3 --
   ArrayCopy(Mas_Ord_Old, Mas_Ord_New);// Saves the preceding history
   Qnt=0;                              // Zeroize orders counter
   ArrayInitialize(Mas_Ord_New,0);     // Zeroize the array
   ArrayInitialize(Mas_Tip,    0);     // Zeroize the array
//------------------------------------------------------------------------------ 4 --
   for(int i=0; i<OrdersTotal(); i++) // For market and pending orders
     {
      if((OrderSelect(i,SELECT_BY_POS)==true)     //If there is the next one
      && (OrderSymbol()==Symbol()))               //.. and our currency pair
        {
         //--------------------------------------------------------------------- 5 --
         Qnt++;                                   // Amount of orders
         Mas_Ord_New[Qnt][1]=OrderOpenPrice();    // Order open price
         Mas_Ord_New[Qnt][2]=OrderStopLoss();     // SL price
         Mas_Ord_New[Qnt][3]=OrderTakeProfit();   // TP price 
         Mas_Ord_New[Qnt][4]=OrderTicket();       // Order number
         Mas_Ord_New[Qnt][5]=OrderLots();         // Amount of lots
         Mas_Tip[OrderType()]++;                  // Amount of orders of the type
         Mas_Ord_New[Qnt][6]=OrderType();         // Order type
         Mas_Ord_New[Qnt][7]=OrderMagicNumber();  // Magic number 
         if (OrderComment()=="")
            Mas_Ord_New[Qnt][8]=0;                // If there is no comment
         else
            Mas_Ord_New[Qnt][8]=1;                // If there is a comment
         //--------------------------------------------------------------------- 6 --
        }
     }
   Mas_Ord_New[0][0]=Qnt;                         // Amount of orders
//------------------------------------------------------------------------------ 7 --
   return;
  }
//------------------------------------------------------------------------------ 8 --
In block 1-2, we give a comment describing the global arrays used in the function. The global arrays are declared in an include file Variables.mqh. In block 3-4, the content of the array Mas_Ord_New is copied to the array Mas_Ord_Old. Thus, the previously known status of orders is stored and can be used further in the program. Then the values of elements of arrays Mas_Ord_New and Mas_Tip showing the new status of orders has been zeroized before the data are updated in block 4-7.

Block 4-7 contains the cycle 'for', in which all market and pending orders are checked one by one for the symbol, to the window of which the EA is attached. The orders are selected using the function OrderSelect() according to the parameter MODE_TRADES set by default. In block 5-6, all required characteristics are calculated for the selected orders, the obtained data are stored in the array of new orders, Mas_Ord_New. At the same time, the amount of orders of all types is calculated, the obtained values being assigned to the corresponding elements of array Mas_Tip. Upon ending of the cycle, the total amount of orders for the symbol is assigned to the element Mas_Ord_New[0][0].

It must be separately noted that closed market orders and deleted pending orders (the execution of the function OrderSelect() with the parameter MODE_HISTORY) are not analyzed. As a rule, the information about closed and deleted orders is not used in trading EAs. The information about closed and deleted orders represent the history of a trading account. This information can be used, for example, to build diagrams that represent the history of capital invested and real trading results. However, it cannot be useful in any way for making new trade decisions. Technically, this part of orders can be accounted in a similar way. However, it is a separate task that has no relation to trading as such.

The events related to orders are analyzed in a program on the basis of comparison of data available in the arrays considered above. For example, if array Mas_Ord_Old contains information about a pending order numbered as 246810, while array Mas_Ord_New contains the data about the same order 246810, but the order is of another type, it means that a pending order has been modified into a market one. It is also necessary to analyze orders when making trades (to be considered later).

Before the function Terminal() is executed for the very first time, arrays Mas_Ord_Old and Mas_Ord_New are empty, i.e., each element of both arrays has zero value. This means that, after the first execution of the function, the array Mas_Ord_Old in the line:

   ArrayCopy(Mas_Ord_Old, Mas_Ord_New);// Store the preceding history
inherits "zero" status from the array Mas_Ord_New, which results in appearance of false event alerts at the execution of the event tracking function. In order to prevent this, the very first execution of the function Terminal() is performed at the stage of initialization, and events are not processed after this execution of the function (see the function init() in Expert Advisor usualexpert.mq4).

Data Function

A normal Expert Advisor used in our practical work favorably compares to its simple analogs, because it provides a trader with high-quality information support.

During trading, the situation changes all the time, various events take place. To make proper decisions, a trader must be fully informed. For this purpose, various functions are used in Expert Advisors. These functions are aimed at informing the user about a certain set of facts and processes.

In simple Expert Advisors, as a rule, this task is performed by the standard function Comment() that displays the preset text in the upper left corner of the symbol window. This information output method is not quite comfortable, since the text can often be superimposed onto the price chart. So this method can only be used in a limited amount of cases, to display short messages.

We will consider here a conceptually different method of displaying information - the entire message is shown in a separate window, whereas graphical objects are used to form message texts. The use of graphical objects produces a tangible advantage, since you can move objects (unlike texts shown using Comment()) creating, in this manner, your message history.

A separate subwindow for displaying of information is created using a custom indicator adjusted correspondingly. The only purpose of this indicator is the creation of that subwindow, so no calculations are performed in the indicator, nor indicating lines are built in it. The code of the indicator Inform.mq4 may look as follows:

//--------------------------------------------------------------------
// Inform.mq4 
// The code should be used for educational purpose only.
//--------------------------------------------------------------------
#property indicator_separate_window // Separate indicator window
//--------------------------------------------------------------------
int start()                         // Special function start()
  {
  }
//--------------------------------------------------------------------
Generally, a programmer may add into the indicator his or her desired code and contribute to its properties. For example, you can display indicating lines in a certain part of the indicator subwindow. In the above example, a simple code of the indicator is given, in the window of which graphical objects will be shown.

User-Defined Function Inform()

int Inform(int Mess_Number,int Number=0,double Value=0.0)
The function displays in the subwindow of the indicator Inform.mq4 messages created using graphical objects. The function controls the position of graphical objects in the indicator subwindow: every new message is displayed in the lower part of the window (lower line) and colored in the desired color, the previously shown messages being moved into the upper part of the window (one line upward). If no new messages are displayed within 15 seconds, all previously displayed messages in the window will be colored in gray (in order not to create diversions for the user) when the program refers to the function.

Parameters:

Mess_Number - message number that can take the following values:

· (zero) 0 - no message is displayed, this mode is used to reset the timer; 

· (minus one) -1 - all graphical objects created by the function will be deleted; 

· (one or more) - the number of the message to be displayed in the indicator subwindow; 

Number - integer used in some messages;

Value - real number used in some messages.

The function Inform() that creates graphical objects, like other functions in a normal EA,is formed as an include file, Inform.mqh:

//----------------------------------------------------------------------------
// Inform.mqh
// The code should be used for educational purpose only.
//----------------------------------------------------------------------- 1 --
// Function that displays graphical messages on the screen.
//----------------------------------------------------------------------- 2 --
int Inform(int Mess_Number, int Number=0, double Value=0.0)
  {
   // int    Mess_Number               // Message number  
   // int    Number                    // Integer to be passed
   // double Value                     // Real number to be passed
   int    Win_ind;                     // Indicator window number
   string Graf_Text;                   // Message line
   color  Color_GT;                    // Color of the message line
   static int    Time_Mess;            // Last publication time of the message
   static int    Nom_Mess_Graf;        // Graphical messages counter
   static string Name_Grf_Txt[30];     // Array of graphical message names
//----------------------------------------------------------------------- 3 --
   Win_ind= WindowFind("inform");      // Searching for indicator window number
   if (Win_ind<0)return;               // If there is no such a window, leave
//----------------------------------------------------------------------- 4 --
   if (Mess_Number==0)                 // This happens at every tick
     {
      if (Time_Mess==0) return;        // If it is gray already
      if (GetTickCount()-Time_Mess>15000)// The color has become updated within 15 sec
        {
         for(int i=0;i<=29; i++)       // Color lines with gray
            ObjectSet( Name_Grf_Txt[i], OBJPROP_COLOR, Gray);
         Time_Mess=0;                  // Flag: All lines are gray
         WindowRedraw();               // Redrawing objects
        }
      return;                          // Exit the function
     }
//----------------------------------------------------------------------- 5 --
   if (Mess_Number==-1)                // This happens at deinit()
     {
      for(i=0; i<=29; i++)             // By object indexes
         ObjectDelete(Name_Grf_Txt[i]);// Deletion of object
      return;                          // Exit the function
     }
//----------------------------------------------------------------------- 6 --
   Nom_Mess_Graf++;                    // Graphical messages counter
   Time_Mess=GetTickCount();           // Last publication time 
   Color_GT=Lime;
//----------------------------------------------------------------------- 7 --
   switch(Mess_Number)                 // Going to message
     {
      case 1:
         Graf_Text="Closed order Buy "+ Number;
         PlaySound("Close_order.wav");                         break;
      case 2:
         Graf_Text="Closed order Sell "+ Number;
         PlaySound("Close_order.wav");                         break;
      case 3:
         Graf_Text="Deleted pending order "+ Number;
         PlaySound("Close_order.wav");                         break;
      case 4:
         Graf_Text="Opened order Buy "+ Number;
         PlaySound("Ok.wav");                                  break;
      case 5:
         Graf_Text="Opened order Sell "+ Number;
         PlaySound("Ok.wav");                                  break;
      case 6:
         Graf_Text="Placed pending order "+ Number;
         PlaySound("Ok.wav");                                  break;
      case 7:
         Graf_Text="Order "+Number+" modified into the market one";
         PlaySound("Transform.wav");                           break;
      case 8:
         Graf_Text="Reopened order "+ Number;                  break;
         PlaySound("Bulk.wav");
      case 9:
         Graf_Text="Partly closed order "+ Number;
         PlaySound("Close_order.wav");                         break;
      case 10:
         Graf_Text="New minimum distance: "+ Number;
         PlaySound("Inform.wav");                              break;
      case 11:
         Graf_Text=" Not enough money for "+
         DoubleToStr(Value,2) + " lots";
         Color_GT=Red;
         PlaySound("Oops.wav");                                break;
      case 12:
         Graf_Text="Trying to close order "+ Number;
         PlaySound("expert.wav");                              break;
      case 13:
         if (Number>0)
            Graf_Text="Trying to open order Sell..";
         else
            Graf_Text="Trying to open order Buy..";
         PlaySound("expert.wav");                              break;
      case 14:
         Graf_Text="Invalid password. EA doesn't function.";
         Color_GT=Red;
         PlaySound("Oops.wav");                                break;
      case 15:
         switch(Number)                 // Going to the error number
           {
            case 2:   Graf_Text="Common error.";                    break;
            case 129: Graf_Text="Wrong price. ";                    break;
            case 135: Graf_Text="Price changed. ";                  break;
            case 136: Graf_Text="No prices. Awaiting a new tick.."; break;
            case 146: Graf_Text="Trading subsystem is busy";        break;
            case 5 :  Graf_Text="Old version of the terminal.";     break;
            case 64:  Graf_Text="Account is blocked.";              break;
            case 133: Graf_Text="Trading is prohibited";            break;
            default:  Graf_Text="Occurred error " + Number;//Other errors
           }
         Color_GT=Red;
         PlaySound("Error.wav");                                    break;
      case 16:
         Graf_Text="Expert Advisor works only for EURUSD";
         Color_GT=Red;
         PlaySound("Oops.wav");                                     break;
      default:
         Graf_Text="default "+ Mess_Number;
         Color_GT=Red;
         PlaySound("Bzrrr.wav");
     }
//----------------------------------------------------------------------- 8 --
   ObjectDelete(Name_Grf_Txt[29]);      // Deleting 29th (upper) object
   for(i=29; i>=1; i--)                 // Cycle for array indexes ..
     {                                 // .. of graphical objects
      Name_Grf_Txt[i]=Name_Grf_Txt[i-1];// Raising objects:
      ObjectSet( Name_Grf_Txt[i], OBJPROP_YDISTANCE, 2+15*i);
     }
   Name_Grf_Txt[0]="Inform_"+Nom_Mess_Graf+"_"+Symbol(); // Object name
   ObjectCreate (Name_Grf_Txt[0],OBJ_LABEL, Win_ind,0,0);// Creating
   ObjectSet    (Name_Grf_Txt[0],OBJPROP_CORNER, 3   );  // Corner
   ObjectSet    (Name_Grf_Txt[0],OBJPROP_XDISTANCE, 450);// Axis Х
   ObjectSet    (Name_Grf_Txt[0],OBJPROP_YDISTANCE, 2);  // Axis Y
   // Текстовое описание объекта
   ObjectSetText(Name_Grf_Txt[0],Graf_Text,10,"Courier New",Color_GT);
   WindowRedraw();                      // Redrawing all objects
   return;
  }
//----------------------------------------------------------------------- 9 --
In block 2-3, the variables used in the function are described. To store the names of graphical objects, the array Name_Grf_Txt is used. According to the method accepted in the function, the program creates a new graphical object for each new message. The total amount of objects is 30, each object represents a text entry located in one line. In case of large screen resolution, the amount of lines to be shown can be increased by increasing of the amount of objects to be created.

In block 3-4, the number of the subwindow of the indicator "Inform" is calculated for messages to be shown in. If the indicator is not attached, the function stops its operations. If there is no window, the messages cannot be displayed, but this does not affect the workability of the EA - all other functions will work in their normal modes, trades will be performed, as well.

In block 4-5, the message colors are analyzed. The function with the parameter Mess_Number=0 is called in the Expert Advisor at every tick (see the function start() in Expert Advisor usualexpert.mq4). If all objects are gray in the preceding history, the function ends its operations. However, if the value of the variable Time_Mess is nonzero, the properties of all objects are changed, namely, all objects will be colored in gray.

If (block 5-6) the parameter Mess_Number=-1 is specified in the function call, all objects previously created at the execution of the function are deleted. This may be necessary when the EA is detached from the symbol window. In this case, according to the common rule, each application program must delete all objects it created during execution (see the function deinit() in the Expert Advisor usualexpert.mq4).

If the control in the program is passed to block 6-7, this means that it is necessary to create a new graphical object with required properties and place it in the lower part of the indicator subwindow (in the lower line; here, the term of "line" is conditional; in fact, the location of graphical objects is determined by the preset coordinates). Each newly created object has its unique name. To create object names, we use the historical number of the message, this is why the messages counter is placed in block 6-7 (at a later stage, the value of the variable Nom_Mess_Graf is used to form a unique name, block 8-9). It is here where the last publication time is accounted and the basic color is established for new messages (green).

Block 7-8 consists of the operator 'switch', inside which the control is passed according to the value of the parameter Mess_Number specified in the function call. In each 'case' of this operator of the variable Graf_Text, a new value is assigned that is the content of the message to be displayed. A special color is set for some messages, for example, red for important messages. All messages are accompanied with sounds that are the execution of the standard function PlaySound() (see Wave Files).

The creation of a new graphical object and replacing of the existing ones are performed in block 8-9. The amount of objects is limited, so one object (the oldest one) is deleted every time when a new message is published. All other existing objects are moved one line upwards. Objects are moved by changing their properties - vertical coordinates. The horizontal coordinates of objects remain unchanged.

After all necessary preparations have been made (all objects are moved one line upwards), a new object with new unique name and with properties partially determined in block 7-8 is created. The type of the graphical object is Text Label. The objects of this type are positioned relative to the symbol window, which allows the user to move price chart arbitrarily, without affecting the positions of messages.

The function Inform() can be called from any location in the program where, implicitly, a text message should be displayed. As a result of long processing, messages are accumulated in the window. The user can view the messages history by resizing the indicator subwindow (for example, by dragging up its upper edge). Optionally, you can also set the window height in such a way that the visible space displays the desired amount of message lines (three-four lines are usually recommended).

Fig. 156. Symbol window. Message in the indicator Subwindow.

It is easy to see that the diversity of messages displayed by the function can be increased. If you are going to expand the program, it is sufficient just to add the new version of 'case' in the operator 'switch' (block 7-8).

Event Tracking Function

Many events take place during trading. A trader can see some of them directly in the symbol window, for example, market price changes or crossed indicator lines. Other events, though they are interesting for a trader, too, are not explicitly shown anywhere. A considerable part of those events can be detected and processed using MQL4.

For example, your dealing center may change the trading conditions shortly before important news are published or when the market becomes very active. In such cases, the spread or the minimum allowed distance for placing of orders and for the requested stop order prices may be increased. If this happens, it is necessary, first, to detect and take the new trading conditions into consideration, and, second, to inform the trader about these changes.

To solve these tasks, you may use the event tracking function in your Expert Advisor.

User-Defined Function Events()

int Events()
The function calculates the changes in the minimum distance required to place orders and their stop orders, as well as the changes in the list of market and pending orders available on the account. To execute the function, you should use the order accounting function Terminal() in your program. The values of the following global arrays are used:

· Mas_Ord_New - the array of characteristics of orders available as of the moment of the function Terminal() execution; 

· Mas_Ord_Old - the array of characteristics of orders available as of the moment of the preceding execution of the function Terminal(). 

The values of the following global variables are used:
- Level_new - the current value of the minimum distance;
- Level_old - the preceding value of the minimum distance.

To display the messages, the function will use the data function Inform(). If the function Inform() is not included in the Expert Advisor, no messages will be shown.

The event tracking function Events() is formed as the include file Events.mqh:

//--------------------------------------------------------------------------------
// Events.mqh
// The code should be used for educational purpose only.
//--------------------------------------------------------------------------- 1 --
// Event tracking function.
// Global variables:
// Level_new            The new value of the minimum distance
// Level_old            The preceding value of the minimum distance
// Mas_Ord_New[31][9]   The last known array of orders
// Mas_Ord_Old[31][9]   The old array of orders
//--------------------------------------------------------------------------- 2 --
int Events()                              // User-defined function
  {
   bool Conc_Nom_Ord;                     // Matching orders in ..
   //.. the old and the new arrays
//--------------------------------------------------------------------------- 3 --
   Level_new=MarketInfo(Symbol(),MODE_STOPLEVEL );// Last known
   if (Level_old!=Level_new)              // New is not the same as old..
     {                                    // it means the condition have been changed
      Level_old=Level_new;                // New "old value"
      Inform(10,Level_new);               // Message: new distance
     }
//--------------------------------------------------------------------------- 4 --
   // Searching for lost, type-changed, partly closed and reopened orders
   for(int old=1;old<=Mas_Ord_Old[0][0];old++)// In the array of old orders
     {                                    // Assuming the..
      Conc_Nom_Ord=false;                 // ..orders don't match
      //--------------------------------------------------------------------- 5 --
      for(int new=1;new<=Mas_Ord_New[0][0];new++)//Cycle for the array ..
        {                                 //..of new orders
         //------------------------------------------------------------------ 6 --
         if (Mas_Ord_Old[old][4]==Mas_Ord_New[new][4])// Matched number 
           {                              // Order type becomes ..
            if (Mas_Ord_New[new][6]!=Mas_Ord_Old[old][6])//.. different
               Inform(7,Mas_Ord_New[new][4]);// Message: modified:)
            Conc_Nom_Ord=true;            // The order is found, ..
            break;                        // ..so exiting ..
           }                              // .. the internal cycle
         //------------------------------------------------------------------ 7 --
                                          // Order number does not match
         if (Mas_Ord_Old[old][7]>0 &&     // MagicNumber matches
            Mas_Ord_Old[old][7]==Mas_Ord_New[new][7])//.. with the old one
           {               //it means it is reopened or partly closed
                                             // If volumes match,.. 
            if (Mas_Ord_Old[old][5]==Mas_Ord_New[new][5])
               Inform(8,Mas_Ord_Old[old][4]);// ..it is reopening
            else                             // Otherwise, it was.. 
               Inform(9,Mas_Ord_Old[old][4]);// ..partly closing
            Conc_Nom_Ord=true;               // The order is found, ..
            break;                           // ..so exiting ..
           }                                 // .. the internal cycle
        }
      //--------------------------------------------------------------------- 8 --
      if (Conc_Nom_Ord==false)               // If we are here,..
        {                                    // ..it means no order found:(
         if (Mas_Ord_Old[old][6]==0)
            Inform(1, Mas_Ord_Old[old][4]);  // Order Buy closed
         if (Mas_Ord_Old[old][6]==1)
            Inform(2, Mas_Ord_Old[old][4]);  // Order Sell closed
         if (Mas_Ord_Old[old][6]> 1)
            Inform(3, Mas_Ord_Old[old][4]);  // Pending order deleted
        }
     }
//--------------------------------------------------------------------------- 9 --
   // Search for new orders
   for(new=1; new<=Mas_Ord_New[0][0]; new++)// In the array of new orders
     {
      if (Mas_Ord_New[new][8]>0)            //This one is not new, but reopened
         continue;                          //..or partly closed
      Conc_Nom_Ord=false;                   // As long as no matches found
      for(old=1; old<=Mas_Ord_Old[0][0]; old++)// Searching for this order 
        {                                   // ..in the array of old orders
         if (Mas_Ord_New[new][4]==Mas_Ord_Old[old][4])//Matched number..
           {                                          //.. of the order
            Conc_Nom_Ord=true;              // The order is found, ..
            break;                          // ..so exiting ..
           }                                // .. the internal cycle
        }
      if (Conc_Nom_Ord==false)              // If no matches found,..
        {                                   // ..the order is new :)
         if (Mas_Ord_New[new][6]==0)
            Inform(4, Mas_Ord_New[new][4]); // Order Buy opened
         if (Mas_Ord_New[new][6]==1)
            Inform(5, Mas_Ord_New[new][4]); // Order Sell opened
         if (Mas_Ord_New[new][6]> 1)
            Inform(6, Mas_Ord_New[new][4]); // Pending order placed
        }
     }
//-------------------------------------------------------------------------- 10 --
   return;
  }
//-------------------------------------------------------------------------- 11 --
Global arrays and variables required for the function execution are described in block 1-2. In block 2-3, variable Conc_Nom_Ord used in the further code for orders analysis is opened.

The function tracks changes of the minimum distance for placing of orders and stop orders. For this, the current value of the minimum distance Level_new is calculated at each execution of the function (block 3-4) and then compared to the preceding one, the value of Level_old (obtained during the preceding execution of the function). If the values of these variables are not equal to each other, it means that the minimum distance has been changed by the dealing center shortly before the last execution of the function. In this case, the current value of the minimum distance is assigned to the variable Level_old (in order to consider it in the subsequent executions of the function), and the function Inform() is executed in order to display the corresponding message.

Generally, you can use a similar method to detect other events, for example, changes in spread, permissions to trade the given symbol (identifier MODE_TRADEALLOWED in the function MarketInfo()), the completion of a new bar (see Problem 27), the fact of crossing indicator lines (see Fig. 107 ), the fact of reaching a certain preset time, etc. The program may detect some events to use the obtained values in your EA, other events - to inform the user about them.

In blocks 4-10, the states of market and pending orders are analyzed. Information about most changes in orders is provided to the user. The analysis is performed in two stages. At the first stage, the program detects changes that relate to lost (closed or deleted), type-changing, partly closed and reopened orders (block 4-9). At the second stage (block 9-10), the new orders are searched for.

In blocks 4-9, the orders accounted in the array Mas_Ord_Old are analyzed. The amount of iterations in the external cycle 'for' is found according to the total amount of orders in the array (array element Mas_Ord_Old[0][0]). To check whether the order is kept as of the current moment, it is necessary to find a similar order in the orders array Mas_Ord_New. This search is performed in the internal cycle 'for' (block 6-8), the amount of iterations of which is equal to the amount of orders in the array (array element Mas_Ord_New[0][0]). We will further name the array Mas_Ord_Old 'old array', whereas the Mas_Ord_New - 'new array'.

In blocks 6-8, the program searches for only those orders, the characteristics of which are different. For example, in block 6-7, the order is checked for its number (see the correspondence of array indexes with order characteristics in Table 4). If the old-array order under check matches in number with one of the orders in the new array, it means that, at least, this order is not closed (or deleted). It is also necessary to check whether the order type is changed. If yes, it means that a pending order is modified into a market one. In this case, the corresponding message is displayed using the function Inform(). Independently on the fact of changing (or keeping unchanged) of the order type, this order will not be analyzed further: the program exits the internal cycle and, finally, starts a new iteration of the external cycle.

If the program finds at the execution of block 6-7 that the old-array order under check does not match in number with any orders from the new array, the control is passed to block 7-8. Here the program checks whether the current order from the new array has a nonzero MagicNumber (all orders opened and placed by the EA have a nonzero MagicNumber). If it has such a MagicNumber and this parameter coincides with the MagicNumber of the order from the old array under check, it means that this order is traded, but has been changed in some way. There are two situations when order number can be changed.

Situation 1. The order is partly closed. You can partly close a market order (not a pending one!) in two stages according to the technology accepted in MT 4. At the first stage, the initial order is completely closed. At the same time, a new market order of a smaller volume is opened with the same open price and with the same requested stop-order prices as in the initial order. This new order gets its unique name, other than the number of the initial order being partly closed.

Situation 2. The order is reopened by the dealing center. Some banks (due to their specific internal accounting rules) forcedly close all market orders at the end of trading day and immediately open market orders of the same type and with the same volume, but at the current price and minus swap. This event doesn't affect the economic results of a trading account in any way. Each newly opened order gets its unique number that doesn't match with the numbers of closed orders.

The difference between the two situations above is in the volumes of new orders: they are different in the first situation and they are unchanged in the second one. This difference is used in block 7-8 to distinguish between orders modified for different reasons. In both cases, the corresponding message is displayed ("the order is partly closed' or 'the order is reopened').

If the program has not detected the matching (block 6-7) or inheriting (block 7-8) of the order in the new array by the completion of the internal cycle, it means that the old-array order under check is closed or deleted. In this case, the control is passed to block 8-9, where one or another message will be displayed, according to the order type. In the above example, three kinds of messages are realized: for order Buy, for order Sell and for pending orders of all types. In a general case, this sequence can be slightly changed (extended) - you can create a group of the corresponding messages for each type of pending orders.

At the second stage, the program considers orders from the new order array (block 9-10). This is made in order to detect newly opened and placed orders. In the external cycle 'for', the program searches in all orders, the information about which is stored in the array of new orders. In order to identify the reopened or partly closed orders, the program uses a simple feature - the availability of comment. When partly closing or reopening an order, the server adds a comment that gives the number of the initial order. The EA above doesn't use comments, so the availability of a comment means that the order under check is not new.

If an order doesn't contain a comment, the program searches for an order with the same number in the old array. If the program finds the order having this number among old orders in the internal cycle 'for', it means that the order isn't new, but was opened before. However, if the number of the order from the new array doesn't match with any orders in the old array, it means that this order is an open market order or a placed pending one. In the lower part of block 9-10, the program calls to function Inform() in order to display the correspondent message, according to the order type.

The use of the considered function Events() turns out to be very helpful in practice. Once having used the function in an EA, the programmer usually uses it in his or her further work. It must be separately noted that the functions Events() and Terminal() are closely interrelated. If you are going to make changes in one of these functions (for example, to use other names for global arrays), you must make the corresponding changes in the other function. If you use comments in orders to realize your trading strategy, you should differently process the inheritance characteristic of the order (block 9-10), namely, you should use string functions to analyze the comment.

The amount of events considered in the function Events() can be highly increased. For example, if you want to completely display all events related to orders, you should add the analysis of order characteristics - changes in the requested stop-order prices and in the requested open prices of pending orders, as well as the closing method (whether the orders are closed as opposite orders or each is closed separately) and the reason for closing/deleting of orders (whether the price has reached the requested stop-order level or the order is closed on the trader's initiative, etc.).

Volume Defining Function

For his or her practical work, a trader needs to be able to regulate the amount of lots for new orders to be opened. It is quite difficult to create a universal function for this purpose, since every trading strategy implies its special volume management. For example, some strategies imply the possibility to work with only one market order, whereas others allow to open new market orders regardless of the existing ones. Strategies based on management of different pending orders are known, too, the simultaneous availability of several market and pending orders being allowed in some cases.

One of the most common methods of volume calculation for newly opened orders (for the strategies that allow only one market order to be opened at a time) is the method of progressive investments. According to this method, the collateral cost of each new order is proportional to the free margin available at the moment of trade. If a market order is closed with profit, the allowed amount of lots for the new order increases. If it is closed with a loss, that amount will be decreased.

In the example below, the user-defined function Lot() is considered that allows you to set the volume for newly opening orders using one of the two alternatives:

Alternative 1. User sets the amount of lots for new orders manually.

Alternative 2. The amount of lots is calculated according to the amount of the money allocated by user. The amount of allocated money is set as percentage of free margin.

User-Defined Function Lot()

bool Lot()
The function calculates the amount of lots for new orders. As a result of the function execution, the value of the global variable Lots_New changes: the amount of lots. The function returns TRUE, if the free margin is sufficient for opening an order with the minimum amount of lots (for the symbol, in the window of which the EA is attached). Otherwise, it returns FALSE.

The function uses the values of the following global variables:

· Lots - volume in lots defined by the user; 

· Percent - the percentage of free margin defined by the user. 

To display message, the function uses the data function Inform(). If the function Inform() is not included in the EA, no messages will be displayed.

The function Lot() that determines the amount of lots is formed as include file Lot.mqh:

//----------------------------------------------------------------------------------
// Lot.mqh
// The code should be used for educational purpose only.
//----------------------------------------------------------------------------- 1 --
// Function calculating the amount of lots.
// Global variables:
// double Lots_New - the amount of lots for new orders (calculated)
// double Lots     - the desired amount of lots defined by the user.
// int Percent     - free margin percentage defined by the user
// Returned values:
// true  - if there is enough money for the minimum volume
// false - if there is no enough money for the minimum volume
//----------------------------------------------------------------------------- 2 --
bool Lot()                                     // User-defined function
  {
   string Symb   =Symbol();                    // Symbol
   double One_Lot=MarketInfo(Symb,MODE_MARGINREQUIRED);//!-lot cost
   double Min_Lot=MarketInfo(Symb,MODE_MINLOT);// Min. amount of lots
   double Step   =MarketInfo(Symb,MODE_LOTSTEP);//Step in volume changing
   double Free   =AccountFreeMargin();         // Free margin
//----------------------------------------------------------------------------- 3 --
   if (Lots>0)                                 // Volume is explicitly set..
     {                                         // ..check it
      double Money=Lots*One_Lot;               // Order cost
      if(Money<=AccountFreeMargin())           // Free margin covers it..
         Lots_New=Lots;                        // ..accept the set one
      else                                     // If free margin is not enough..
         Lots_New=MathFloor(Free/One_Lot/Step)*Step;// Calculate lots
     }
//----------------------------------------------------------------------------- 4 --
   else                                        // If volume is not preset
     {                                         // ..take percentage
      if (Percent > 100)                       // Preset, but incorrectly ..
         Percent=100;                          // .. then no more than 100
      if (Percent==0)                          // If 0 is preset ..
         Lots_New=Min_Lot;                     // ..then the min. lot
      else                                     // Desired amount of lots:
         Lots_New=MathFloor(Free*Percent/100/One_Lot/Step)*Step;//Calc
     }
//----------------------------------------------------------------------------- 5 --
   if (Lots_New < Min_Lot)                     // If it is less than allowed..
      Lots_New=Min_Lot;                        // .. then minimum
   if (Lots_New*One_Lot > AccountFreeMargin()) // It isn't enough even..
     {                                         // ..for the min. lot:(
      Inform(11,0,Min_Lot);                    // Message..
      return(false);                           // ..and exit 
     }
   return(true);                               // Exit user-defined function
  }
//----------------------------------------------------------------------------- 6 --
The function has a simple code. In block 1-2, global variables and returned values are described. In block 2-3, the values of some variables are calculated. For calculations, the following priority in setting of values is accepted: If a user has set a non-zero amount of lots, the value of the percentage of free margin is not taken into consideration. External variables Lots and Percent are declared in the include file Variables.mqh.

In block 3-4, the calculations are made for the situation where the user has defined a non-zero value of the volume in lots in the external variable Lots. In this case, the program makes a check. If the free margin is sufficient to open a market order with the defined amount of lots, then the value set by the user will be assigned to the global variable Lots_New and used in further calculations. If the free margin doesn't cover this amount, then the maximum possible amount of lots is calculated that is used further (see Mathematical Functions).

The control is passed to block 4-5, if the user has defined zero amount of lots. At the same time, we take into consideration the percentage of free margin specified by the user in the external variable Percent. The program makes a check: If the value exceeds one hundred (percent), the value of 100 is used in calculations. If the user has defined zero value of the variable Percent, the amount of lots is equated with the minimum possible value set by the dealing center. For all intermediate Для всех промежуточных величин высчитывается количество лотов, соответствующее количеству выделенных пользователем средств.

In block 5-6, the necessary checks are made. If the calculated amount of lots turns out to be less than the minimum allowed one (for example, zero value can be obtained in block 4-5, if the user has defined a small value of the variable Percent), then the minimum value will be assigned to the variable Lots_New. Then the program checks whether there are enough free assets to open an order with the volume of the previously calculated amount of lots (there can be insufficient money on the account). If the money available is not enough, the program displays a message for the user and exits the function, the function returning 'false'. However, the successful check results in returning of 'true'.

Function Defining Trading Criteria

The success of any trading strategy mainly depends on the sequence of trading criteria calculations. The function that defines trading criteria is the most important part of a program and must be used without fail. According to trading strategy, the function may return values that correspond with particular trading criteria.

In a general case, the following criteria can be defined:

· criterion for opening of a market order; 

· criterion for closing of a market order; 

· criterion for partly closing of a market order; 

· criterion for closing of opposite market orders; 

· criterion for modification of the requested prices of stops of a market order; 

· criterion for placing of a pending order; 

· criterion for deletion of a pending order; 

· criterion for modification of the requested open price of a pending order; 

· criterion for modification of the requested prices of stops of a pending order. 

In most cases, the triggering of one trading criterion is exclusive as related to other trading criteria. For example, if the criterion for opening a Buy order becomes important at a certain moment, this means that the criteria used for closing Buy orders or for opening Sell orders cannot be important at the same moment (see Relation of Trading Criteria). At the same time, according to the rules inherent in a given trading strategy, some criteria may trigger simultaneously. For example, the criteria for closing a market order Sell and for modification of a pending order BuyStop may become important simultaneously.

A trading strategy imposes requirements to the content and the usage technology of the function defining trading criteria. Any function can return only one value. So, if you have realized in your Expert Advisor a trading strategy that implies using only mutually exclusive trading criteria, the value returned by the function can be associated with one of the criteria. However, if your strategy allows triggering of several criteria at a time, their values must be passed to other functions for being processed, using global variables for that.

Trading strategy realized in the EA below implies using only mutually exclusive criteria. This is why the function Criterion() for passing the above criteria to other functions uses the value returned by the function.

User-Defined Function Criterion()

int Criterion()
The function calculates trading criteria. It can return the following values:
10 - triggered a trading criterion for closing of market order Buy;
20 - triggered a trading criterion for opening of market order Sell; 
11 - triggered a trading criterion for closing of market order Buy;
21 - triggered a trading criterion for opening of market order Sell;
0 - no important criteria available;
-1 - the symbol used is not EURUSD.

The function uses the values of the following external variables:
St_min - the lower level of indicator Stochastic Oscillator; 
St_max - the upper level of indicator Stochastic Oscillator; 
Open_Level - the level of indicator MACD (for order opening);
Close_Level - the level of indicator MACD (for order closing).

In order to display messages, the function uses the data function Inform().If the function Inform() is not included in the EA, no messages will be displayed.

Function defining trading criteria, Criterion(), is formed as include file Criterion.mqh:

//-------------------------------------------------------------------------
// Criterion.mqh
// The code should be used for educational purpose only.
//-------------------------------------------------------------------- 1 --
// Function calculating trading criteria.
// Returned values:
// 10 - opening Buy  
// 20 - opening Sell 
// 11 - closing Buy
// 21 - closing Sell
// 0  - no important criteria available
// -1 - another symbol is used
//-------------------------------------------------------------------- 2 --
// External variables:
extern int St_min=30;                  // Minimum stochastic level
extern int St_max=70;                  // Maximum stochastic level
extern double Open_Level =5;           // MACD level for opening (+/-)
extern double Close_Level=4;           // MACD level for closing (+/-)
//-------------------------------------------------------------------- 3 --
int Criterion()                        // User-defined function
  {
   string Sym="EURUSD";
   if (Sym!=Symbol())                  // If it is a wrong symbol
     {
      Inform(16);                      // Messaging..
      return(-1);                      // .. and exiting
     }
   double
   M_0, M_1,                           // Value MAIN at bars 0 and 1
   S_0, S_1,                           // Value SIGNAL at bars 0 and 1
   St_M_0, St_M_1,                     // Value MAIN at bars 0 and 1
   St_S_0, St_S_1;                     // Value SIGNAL at bars 0 and 1
   double Opn=Open_Level*Point;        // Opening level of MACD (points)
   double Cls=Close_Level*Point;       // Closing level of MACD (points)
//-------------------------------------------------------------------- 4 --
   // Parameters of technical indicators:
   M_0=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_MAIN,0); // 0 bar
   M_1=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_MAIN,1); // 1 bar
   S_0=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_SIGNAL,0);//0 bar
   S_1=iMACD(Sym,PERIOD_H1,12,26,9,PRICE_CLOSE,MODE_SIGNAL,1);//1 bar
 
   St_M_0=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_MAIN,  0);
   St_M_1=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_MAIN,  1);
   St_S_0=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_SIGNAL,0);
   St_S_1=iStochastic(Sym,PERIOD_M15,5,3,3,MODE_SMA,0,MODE_SIGNAL,1);
//-------------------------------------------------------------------- 5 --
   // Calculation of trading criteria
   if(M_0>S_0 && -M_0>Opn && St_M_0>St_S_0 && St_S_0<St_min)
      return(10);                      // Opening Buy    
   if(M_0<S_0 &&  M_0>Opn && St_M_0<St_S_0 && St_S_0>St_max)
      return(20);                      // Opening Sell 
   if(M_0<S_0 &&  M_0>Cls && St_M_0<St_S_0 && St_S_0>St_max)
      return(11);                      // Closing Buy    
   if(M_0>S_0 && -M_0>Cls && St_M_0>St_S_0 && St_S_0>St_min)
      return(21);                      // Closing Sell         
//-------------------------------------------------------------------- 6 --
   return(0);                          // Exit the user-defined function
  }
//-------------------------------------------------------------------- 7 --
In block 1-2, the values returned by the function are described. In block 2-3, some external variables are declared. The include file Criterion.mqh is the only file used in the considered EA, in which the global (in this case, external) variables are declared. In the section named Structure of a Normal Program, you can find the reasoning for declaring of all global variables without exception in a separate file Variables.mqh. In this case, the external variables are declared in the file Criterion.mqh for two reasons: first, to demonstrate that it is technically possible (it is not always possible); second, to show how to use external variables at debugging/ testing of a program.

It is technically possible to declare external variables in the file Criterion.mqh, because these variables are not used in any other functions of the program. The values of external variables declared in block 2-3 determine the levels for indicators Stochastic Oscillator and MACD and are used only in the considered function Criterion(). The declaration of external variables in the file containing the function that defines trading criteria may be reasonable, if the file is used temporarily, namely, during the program debugging and calculating of optimal values of those external variables. For this purpose, you can add other external variables in the program, for example, to optimize inputs of indicators (in this case, the constants set the values of 12,26,9 for MACD and 5,3,3 for Stochastic Oscillator). Once having finished coding, you may delete these external variables from the program and replace them with constants with the values calculated during optimization.

In block 3-4, the local variables are opened and described. The Expert Advisor is intended to be used on symbol EURUSD, so the necessary check is made in block 3-4. If the EA is launched in the window of another symbol, the function finishes operating and returns the value of -1 (wrong symbol).

In the program, the values of two indicators calculated on the current and on the preceding bar are used (block 4-5). Usually, when you use indicators Stochastic Oscillator and MACD, the signals for buying or selling are formed when two indicator lines meet each other. In this case, we use two indicators simultaneously to define trading criteria. The probability of simultaneous intersection of indicator lines of two indicators is rather low. It is much more probable that they will intersect one by one - first in one indicator, a bit later - in another one. If the indicator lines intersect within a short period of time, two indicators can be considered to have formed a trading criterion.

For example, below is shown how a trading criterion for buying is calculated (block 5-6):

if(M_0>S_0 && -M_0>Opn && St_M_0>St_S_0 && St_S_0<St_min)
According to this record, the criterion for buying is important if the following conditions are met:

· in indicator MACD, indicator line MAIN (histogram) is above indicator line SIGNAL and below the lowest level Open_Level (Fig. 157); 

· in indicator Stochastic Oscillator, indicator line MAIN (histogram) is above indicator line SIGNAL and below the lowest level St_min (Fig. 158). 

[image: image2.png]EURUSD,H1

k0

MACD(12,269) 000070 00054
hr) S : der
) y

Close_Level!

th i te

Ty tite

i}

Infor H

2O z06 12 0n 060 120a 1000 120w 140 120 500 12 0a 220 13 00200

T oatenn 13 0a 100





Рис. 157. Necessary condition of MACD indicator line positions to confirm the importance of trading criteria for opening and closing of orders.

In the left part of Fig. 157, the positions of MACD indicator lines is shown, at which two criteria trigger - opening of Buy and closing of Sell. Indicator line MAIN is below the level of 0.0005 within the time of T1 = t 1 - t 0. If the necessary indications of Stochastic Oscillator occur at this moment, the criterion for opening of Buy will trigger. Line MAIN is below the level of 0.0004 within the time T2 = t 2 - t 0. If the indications of Stochastic Oscillator confirm this position, the criterion for closing of Sell will trigger.

Please note that, formally, both criteria above trigger within T1 (if confirmed by Stochastic Oscillator). It was mentioned before that the considered function Criterion() returns only one value, namely, the value assigned to one triggered criterion. During this period, it becomes necessary to choose one of the criteria. This problem is solved in advance, during programming, according to the priorities prescribed by the trading strategy.

In this case (according to the trading strategy considered), the priority of opening a Buy order is higher than that of closing a Sell order. This is why, in block 5-6, the program line, in which the criterion for opening of Buy, is positioned above. If during the period of T1 (Fig. 157), we have got the confirmation from Stochastic Oscillator, the function returns 10 assigned to this criterion. Within the period from t1 to t2, the function will return 21 assigned to the Sell closing criterion.

At the same time, at the execution of trade functions, the necessary trade requests will be formed. At triggering of the criterion for opening of Buy, first of all, the trade requests for closing of all available Sell orders will be formed. As soon as no such orders are left, opening of one Buy order will be requested. Respectively, when the criterion for closing of Sell orders triggers, a sequence of trade requests for only closing of all Sell orders will be formed (see Trade Functions).

The conditions, at which Stochastic Oscillator confirmation triggers, are shown in Fig. 158.

Fig. 158. Necessary condition of Stochastic Oscillator indicator line positions to confirm the importance of trading criteria for opening and closing of orders.

According to the program code specified in block 5-6, the criteria for opening of Buy and closing of Sell can become important provided the indicator line MAIN turns out to be above the signal line SIGNAL in Stochastic Oscillator, line MAIN being below the minimum level St_min. In Fig. 158, such conditions are formed within the period of Ts. The mirrored conditions confirm the triggering of criteria for opening of the order Sell and closing of the order Buy (in the right part of Fig. 158). If no criterion has triggered, the function returns 0. Other trades can be made under these conditions, for example, correction of the stop levels requested.

It must be noted separately that the considered trading strategy implies the usage of indications produced by MACD calculated on the one-hour timeframe, whereas Stochastic Oscillator is calculated on the 15-minute timeframe.The timeframe may be changed during testing in order to optimize the strategy. However, after testing, in the final code of the function Criterion(), it is necessary to specify constant value for all parameters calculated, including timeframes. The EA must be used only under the conditions, for which it has been created. In the example above (with the values of PERIOD_H1 and PERIOD_M15 specified explicitly in the indicators), the EA will consider only necessary parameters regardless the current timeframe set in the symbol window, where the EA has been launched.

	Trading criteria accepted in this given EA are used for training purposes and should not be considered as a guide to operations when trading on a real account.

